Gallery of functions

This plot show the functions x, x^2, x^3, x^4 and x^5. Note that the higher powers get more and more strongly
curved. When $x < 0$, the odd powers of x (ie x, x^3 and x^5) are negative, but the even powers (x^2 and x^4)
are positive. When $x>0$, all the powers are positive.

```
> plot([seq(x^k,k=1..5)],x=-1..1,scaling=constrained);
```



The next picture shows the function

$$f(x) = \prod_{k=-3}^{3} \left(x - \frac{k}{3}\right) = (x-1)\left(x-\frac{2}{3}\right)\left(x-\frac{1}{3}\right)x\left(x+\frac{1}{3}\right)\left(x+\frac{2}{3}\right)(x+1)$$

This has seven roots spread out evenly between $x = -1$ and $x = 1$, so it stays quite close to zero in that interval. Outside that interval, however, it gets very large very quickly.

```
> plot(product(x-k/3,k=-3..3),x=-2..2,y=-1..1);
```



The next picture shows the function $f(x) = \frac{x^8}{576} - \frac{5\,x^6}{96} + \frac{91\,x^4}{192} - \frac{205\,x^2}{144} + 1$

This is specially cooked up so that $f(0) = 1$ and f(-4) = f(-3) = f(-2) = f(-1) = f(1) = f(2) = f(3) = f(4) = 0. It is in fact the *unique* polynomial of degree 8 with these properties.

It is a common mistake to think that a function cooked up in this way should stay reasonably close to zero between $x = -4$ and $x = -1$, and between $x = 1$ and $x = 4$. You can see from the graph that it actually swings around quite wildly.

```
> f := unapply(expand(product(x^2-k^2,k=1..4)/576),x);
```

$$f := x \rightarrow \frac{1}{576}\,x^8 - \frac{5}{96}\,x^6 + \frac{91}{192}\,x^4 - \frac{205}{144}\,x^2 + 1 \quad \textbf{(1)}$$

```
> plot(f(x),x=-4.1..4.1,xtickmarks=9);
```



The next graph shows a function of the form $y = \frac{a\,x + b}{c\,x + d}$; such functions are called *Mobius transformations*.

It is an important fact that any such function can be inverted: we have $x = \frac{d\,y - b}{-c\,y + a}$. Invertibility means that
the graph crosses every horizontal line in exactly one place.

```
> plot((2*x+3)/(3*x-4),x=-5..5,y=-5..5,discont=true);
```



The next two pictures show the functions $y = \frac{x}{\sqrt{1-x^2}}$ and $x = \frac{y}{\sqrt{1+y^2}}$, which are inverse to each other. The variable x runs from -1 to +1 (not including the endpoints), and the variable y runs over the whole real line (although we have only shown the interval from -3 to +3).

```
> plot(x/sqrt(1-x^2),x=-1..1,y=-3..3,scaling=constrained);
```




```
> plot(y/sqrt(1+y^2),y=-3..3,x=-1..1,scaling=constrained);
```



The next picture shows the functions $y = \mathrm{e}^x$ (in red) and $y = \log(x)$ (in green). Note that $\log(x)$ is not defined when
$x < 0$, so there is no green curve in the left hand half of the picture. Note also that the two functions are inverse to
each other, in the sense that $\log(\mathrm{e}^x) = x$ (for all x) and $\mathrm{e}^{\log(x)} = x$ (for all x>0). This means that the green curve
is obtained from the red one by reflecting in the line $x = y$.

```
> plot([exp(x),log(x)],x=-4..4,y=-4..4);
```



The next picture shows an exponential decay function $y = \mathrm{e}^{-t}$. Functions like this occur often in probability.
For example, if you are watching a series of random events (such as thunderstorms, people joining a queue, or whatever) then the probability that you have to wait for a time t before the next event is often given by e^{-t}. (More precisely, it is given by $\mathrm{e}^{-a\,t}$ for some constant a, but the value of the constant affects only the size of the graph and not its overall shape. Many of our examples will implicitly depend on arbitrary constants like this, and we will generally ignore them.)

```
> f := t -> exp(-t):
> plot(f(t),t=(-0.5)..5);
```



The next example is a decaying oscillation, of the form $f(t) = a\,e^{-b\,t}\sin(c\,t)$. If you pluck a guitar string at time 0,
then the displacement at time t varies according to a function like this.

```
> f := t -> 2 * exp(-t) * sin(30*t):
> plot( f(t), t=0..5);
```



The curve below, with equation $y = e^{x} + e^{-x}$, is called a *catenary*. It is the shape adopted by a length of chain that is attached at the ends, but otherwise hangs freely. (This is proved by the Calculus of Variations, which is covered in AMA314 (Optimal Control Theory)).

```
> plot(exp(x) + exp(-x),x=-1..1);
```



Functions like $x^n\,\mathrm{e}^{-x}$ occur very frequently in statistical physics. Below we have plotted the graph for $n=6$,
but you should really think of n as being much larger, like the number of molecules of gas in some physical system,
which could easily by 10^{25} or so.

When x is small, the function grows very quickly, like x^n does. However, when we get past $x=10$ or so,
the factor e^{-x} gets extremely small, so quickly that the effect of the large x^n term is wiped out, and the graph soon becomes indistinguishable from zero.

```
> plot(x^6*exp(-x),x=0..50);
```



The picture below shows the function $y = e^{-x^2}$, which is of fundamental importance in statistics.

```
> plot(exp(-x^2),x=-3..3,scaling = constrained);
```



The next example shows a function of the form $e^{-x^2}\sin(b\,x)$. It oscillates rapidly with amplitude decaying rapidly to zero as x moves outside an interval of length about 4 centred at the origin. In quantum
mechanics one often sees functions like this, with the position of the bump moving along the x axis over
time; they are called *wave packets.*

```
> plot(exp(-x^2) * sin(40*x),x=-3..3);
```



We now plot the function $y = e^{-\frac{1}{T}}$, which occurs frequently in thermodynamics. You should think of T as temperature, measured in a suitable scale where $T = 0$ corresponds to absolute zero (approximately
-273 degrees centigrade). The key point about the graph below is that it is extremely flat at the point $T = 0$,
and it remains extremely flat no matter how much you zoom in and magnify the vertical scale. This is a
mathematical reflection of the strange and extreme behaviour of absolute zero.

```
> plot(exp(-1/T),T=0..2);
```



The picture below shows the function $y = \cos(x^2)$. When x is reasonable large, the angle x^2 varies very rapidly, so $\cos(x^2)$ oscillates wildly between +1 and -1. This mean that in any short interval, the positive
values will cancel out the negative values and the average value will be close to zero.
Near the origin, however, the angle x^2 changes quite slowly, so $\cos(x^2)$ will generally not change sign in
a short interval, and the average value over a short interval will be nonzero.

This is known as the *principle of stationary phase;* it is very important in quantum mechanics.

```
> plot(cos(x^2),x=-10..10);
```



The next picture shows a function of the form $a \sin(u\,t) + b \sin(v\,t)$, where the two frequencies u and v
are very close together. The result is a very rapid oscillation (of frequency equal to the average of u and v)
whose overall size varies much more slowly (at frequency $\frac{u-v}{2}$).

This is relevant to the process of tuning up an orchestra. Two players will play what is nominally the same
note on their violins, generating signals of frequency u and v, where u and v are supposed to be the same.
If they are not in fact the same, then they will hear a "beat" of frequency $\frac{u-v}{2}$, and they will adjust one or
the other instrument until the beat goes away.

```
> plot(2*sin(10*t) + 3*sin(10.5*t),t=-30..30,numpoints=1000);
```



The next graph shows function of the form $a\,t + b\sin(u\,t)$, representing an oscillating signal superimposed on a steady drift. The graph of atmospheric carbon dioxide concentration over the last century or so looks very much like this: a yearly oscillation caused by seasonal effects, plus a steady increase presumably caused by the burning of fossil fuels.

```
> plot(.9 * t+.5 * sin(10*t),t=-10..10);
```



The next picture shows a function of the form $\sin((u+\sin(v\,t))\,t)$, where u is supposed to be much larger than v (and also much larger than 1). This function is essentially an oscillation of frequency u, except that this frequency itself changes at the much lower frequency v. This is the kind of signal produced by an FM (= frequency modulated) radio transmitter. In that context, u is a radio frequency (perhaps 10^8 or so) and v is an audio frequency (perhaps 10^4). The picture is not very good, and is included as a reminder of the limitations of computer graphics. Pictures can help, but you need to understand the formulae as well.

```
> plot(sin((6 + sin(t)) * t),t=-20..20,numpoints=1000);
```



The next picture shows a square wave of frequency 2. This is like the timing signal generated by the clock circuit of a computer chip (except that in that case the frequency would be more like $2\ 10^9$, for a 2 GHz chip).

```
> plot((-1)^floor(2*x),x=-5..5,scaling=constrained);
```



The next picture shows that you can build an approximate square wave out of sin waves of different frequencies. This is the first inkling of Fourier Theory, which will be studied in PMA212 (Linear Mathematics II).
Fourier theory is the mathematical basis for musical synthesisers, among many other things.

```
> plot(sin(x) + sin(3*x)/3 + sin(5*x)/5 + sin(7*x)/7,x=-2*Pi..2*Pi)
  ;
```




```
> plot((1-v^2)^(-1/2),v=-1..1,g=0..4);
```




```
> ?spline
> g := unapply(spline([-4,-3,-2,-1,0,1,2,3,4],[0,0,0,0,1,0,0,0,0],
  x,quadratic),x):
> g(1.1);
```

$$-0.058921568 \tag{2}$$

```
> plot(g(x),x=-4.1..4.1);
```


1
0.8
0.6
0.4
0.2
−4
−3
−2
−1
0
1
2
3
4
x